```
# Wave field extrapolation program
implicit undefined (a-z)
complex cd(48),ce(48),cf(48),q(48),aa,a,b,c,cshift
real p(96,48,12),phase,pi2,dx,dz,v,z0,x0,dt,dw,lambda,w,wov,x
integer ix,nx,iz,nz,iw,nw,it,nt
open(3,file='plot30',status='new',access='direct',form='unformatted',recl=1)

nt=12; nx=48; nz=96; dx=2; dz=1; pi2=2.*3.141592
v=1; lambda=nz*dz/4; dw=v*pi2/lambda; dt=pi2/(nt*dw); nw=2

do iz=1,nz;  do ix=1,nx;  do it=1,nt  { p(iz,ix,it) = 0. }
do iw = 1,nw {                          # superimpose nw frequencies
        w =  iw*dw;      wov = w/v              # frequency / velocity
        x0 = nx*dx/3; z0 = nz*dz/3
        do ix = 1,nx {                         # initial conditions for a
                x = ix*dx-x0;                  # collapsing spherical wave
                phase = -wov*sqrt(z0**2+x**2)
                q(ix) = cexp(cmplx(0.,phase))
                }
        aa = dz/(4.*(0.,-1.)*wov*dx**2)        # tridiagonal matrix coefficients
        a = -aa;      b = 1.+2.*aa;      c = -aa
        do iz = 1,nz {                         # extrapolation in depth
                do ix = 2,nx-1          # diffraction term
                        cd(ix) = aa*q(ix+1) + (1.-2.*aa)*q(ix) + aa*q(ix-1)
                cd(1) = 0.;      cd(nx) = 0.
                call ctris(nx,-a,a,b,c,-c,cd,q,ce,cf)
                        # "ctris" solves complex tridiagonal equations
                        # i.e. "rtris" with complex variables
                cshift = cexp(cmplx(0.,wov*dz))
                do ix = 1,nx                   # shifting term
                        q(ix) = q(ix) * cshift
                do it=1,nt {                   # evolution in time
                        cshift = cexp(cmplx(0.,-w*it*dt))
                        do ix = 1,nx
                                p(iz,ix,it) = p(iz,ix,it)+q(ix)*cshift
                        }
                }
        }
write(3,rec=1) (((p(iz,ix,it),iz=1,nz),ix=1,nx),it=1,nt)
stop;           end
```